

An efficient, provably exact algorithm for the 0-1 loss linear classification problem

Xi He¹, Max A. Little^{1,2*}

June 22, 2023

Abstract

Algorithms for solving the linear classification problem have a long history, dating back at least to 1936 with linear discriminant analysis. For linearly separable data, many algorithms can obtain the exact solution to the corresponding 0-1 loss classification problem efficiently, but for data which is not linearly separable, it has been shown that this problem, in full generality, is NP-hard. Alternative approaches all involve approximations of some kind, including the use of surrogates for the 0-1 loss (for example, the hinge or logistic loss) or approximate combinatorial search, none of which can be guaranteed to solve the problem exactly. Finding efficient algorithms to obtain an exact i.e. globally optimal solution for the 0-1 loss linear classification problem with fixed dimension, remains an open problem. In research we report here, we detail the construction of a new algorithm, incremental cell enumeration (ICE), that can solve the 0-1 loss classification problem exactly in polynomial time. To our knowledge, this is the first, rigorously-proven polynomial time algorithm for this long-standing problem.

1 Introduction

There has been an increasing trend to leverage machine learning (ML) for high-stakes prediction applications that deeply impact human lives. Many of these ML models are “black boxes” with highly complex, inscrutable functional forms. In high-stakes applications such as healthcare and criminal justice, black box ML predictions have incorrectly denied parole [Wexler, 2017], misclassified highly polluted air as safe to breathe [McGough, 2018], and suggested poor allocation of valuable, limited resources in medicine and energy reliability [Varshney and Alemzadeh, 2017]. In such high-stakes applications of ML, we always want the best possible prediction, and we want to know how the model makes these predictions so that we can be confident the predictions are meaningful [Rudin, 2022]. In short, the ideal model is simple enough to be easily understood (*interpretable*), and optimally accurate (*exact*).

Hence, in high-impact applications of ML, we always want the best possible prediction, and we want to know how the model makes these predictions so that we can be confident the predictions are meaningful. In short, particularly in high-stakes applications of ML, the ideal model is simple enough to understand and optimally accurate, then our interpretations of the results can be faithful to what our model actually computes.

Linear classification algorithms are widely used because they are simple and intuitive. The 0-1 loss objective (minimizing the number of misclassifications) for the binary linear classifier, is known to lead to models which are robust to outliers, but hard to optimize directly. To illustrate, assume a data set consists of N data points (or data items) \mathbf{x}_n , $\forall n \in \{1, \dots, N\} = \mathcal{N}$, where the data points $\mathbf{x}_n \in \mathbb{R}^D$ and D is the dimension of the *feature space*. Each data point has a unique true label $l_n \in \{-1, 1\}$, $\forall n \in \mathcal{N}$. All true labels in this data set are stored in a set $\mathbf{l} = \{l_1, l_2, \dots, l_N\}^T$. The data points and their labels are packaged together into the dataset \mathcal{D} . The objective function for 0-1 loss linear classification problem can be defined as

$$E_{0-1}(\mathbf{w}) = \sum_{n \in \mathcal{N}} \mathbf{1}[\text{sign}(\mathbf{w}^T \bar{\mathbf{x}}_n) \neq l_n], \quad (1)$$

which is a sum of 0-1 loss functions $\mathbf{1}[\cdot]$, each taking the value 1 if the Boolean argument is true, and 0 if false. The function sign returns +1 if the argument is positive, and -1 if negative (and zero otherwise). The linear decision function $\mathbf{w}^T \bar{\mathbf{x}}$ with parameters $\mathbf{w} \in \mathbb{R}^{D+1}$ ($\bar{\mathbf{x}}$ is the data in homogeneous coordinates) is highly interpretable since it represents a simple hyperplane boundary in feature space separating the two classes. It would therefore be extremely useful to find an

^{*}1School of Computer Science, University of Birmingham, Birmingham, UK, ²Media Lab, Massachusetts Institute of Technology, Cambridge, MA, USA

optimal linear decision function for a given dataset \mathcal{D} . Equation (1) counts the number of misclassified data points given the parameter \mathbf{w} , so that the supervised classification problem is solved by computing

$$\hat{\mathbf{w}} = \underset{\mathbf{w}:\mathbb{R}^{D+1}}{\operatorname{argmin}} E_{0-1}(\mathbf{w}). \quad (2)$$

Although apparently simple, this is a surprisingly challenging optimization problem. Considered a continuous optimization problem, the standard ML optimization technique, gradient descent, is not applicable (since the gradients of $E_{0-1}(\mathbf{w})$ with respect to \mathbf{w} are zero everywhere they exist), and the problem is non-convex so there are a potentially very large number of local minima in which gradient descent can become trapped. Heuristics exist, in particular the classic *perceptron training algorithm* and variants which are only guaranteed to find one of these local minima. By replacing the loss function $\mathbf{1}[\cdot]$ with more manageable *surrogates* such as the *hinge loss* $E_{\text{hinge}}(\mathbf{w}) = \sum_{n \in \mathcal{N}} \max(0, 1 - l_n \mathbf{w}^T \mathbf{x}_n)$ which is convex and differentiable nearly everywhere, the corresponding optimization is a linear problem solvable by general-purpose algorithms such as *interior-point primal-dual* optimization. However, this can only find a sub-optimal decision function corresponding to an upper bound on the globally optimal value of the objective $E_{0-1}(\mathbf{w})$.

An alternative is to recast the problem as a combinatorial one, in the following way. Every choice of parameters \mathbf{w} determines a particular classification prediction or *assignment*, $\operatorname{sign}(\mathbf{w}^T \mathbf{x}_n) \in \mathcal{S}$ for all $n \in \mathcal{N}$, where \mathcal{S} is the *discrete search* or *solution space*, in linear classification problem \mathcal{S} is all possible 2^N *assignments/configurations*. Every assignment entails a particular total loss $E_{0-1}(\mathbf{w})$. For finite N there are only a finite number of equivalence classes of parameters \mathbf{w} which entail the same assignment $\mathbf{z} = \{z_1, z_2, \dots, z_N\} \in \mathcal{S}$, that is, the set $\{\mathbf{w} \in \mathbb{R}^{D+1} : \operatorname{sign}(\mathbf{w}^T \mathbf{x}_n) = z_n, \forall n \in \mathcal{N}\}$. In fact, the geometry of the problem implies that not all assignments correspond to one of these equivalence classes; the only ones which do are known as *dichotomies* or *half spaces*, and while there are 2^N possible assignments, there are only $O(N^D)$ dichotomies for data set in D dimensions [Cover, 1965]. Thus, problem (2) can instead be treated as a combinatorial optimization problem of finding the best such dichotomies and their implied assignments, from which an optimal parameter $\hat{\mathbf{w}}$ can be obtained.

This is still a challenging optimization problem, and indeed, it has been known as NP-hard [Ben-David et al., 2003, Feldman et al., 2012]. Because this problem involves both continuous-valued parameters \mathbf{w} and discrete-valued assignments \mathbf{z} , it is an example of a *mixed continuous-discrete optimization problem* (more precisely in this case, a *mixed integer-linear programming*, MIP problem) for which general-purpose algorithms such as *branch-and-bound* (BnB) or *mixed integer programming solvers* (GLPK for instance) are applicable. Nonetheless, such generic algorithms do not come with guarantees on computational complexity, and are usually worst-case exponential time as in principle they will test all 2^N possible assignments.

In our research here, we will construct an algorithm to solve (2) exactly, with polynomial-time guarantees on worst-case computational complexity. The proof of the correctness of this algorithm i.e. that it solves (2), relies on several ideas in computational geometry and oriented matroid theory. We take a fundamentally different approach, based on several modern, broadly-applicable algorithm design principles such as sequential decision processes, short-cut filter fusion, tupling, and other acceleration techniques. We shall use functional programming to derive an efficient algorithm from provably correct specification.

The paper is organized as follows. In Section 2, we explain in detail how our algorithm is constructed. Section 3 shows the results of empirical comparisons against approximate algorithms applied to classification data sets from the UCI machine learning repository. Finally, we present a summary of the work, a brief discussion and suggest future research direction in Section 4.

2 Solving the problem: functional program derivation meets combinatorial geometry

It is widespread practice for machine learning algorithms that solve problems like (2) to be given with little justification. Where formal proofs of correctness—essential to *guaranteeing exactness* in our case—are provided, these proofs are given “after the fact”. While this practice is not necessarily problematic, it can be error-prone and *ambiguous*, and provide little reason as to *how* the algorithm was developed. We argue that it is preferable to *derive* algorithms from a *correct specification*, using simple, calculational steps, a process known as *program calculation* [Bird and de Moor, 1997]. This requires treating programs as if they are mathematical functions, and it is for this reason we develop our algorithm in a *strongly-typed* and *side-effect free functional programming* language, specifically Haskell. Strong typing provides the substantial benefit that algorithm correctness requires *type correctness* of the program (and up to a certain degree of freedom, the implication goes the other way as well), and the syntactic and semantic rules of the computer language avoid ambiguity. For readers unfamiliar with this language and functional programming in general, therefore, we will give brief

tutorial notes alongside the derivations. All the code in the paper is directly executable Haskell so that the reader can test out the ideas by direct experimentation if desired.

2.1 Generic sequential decision processes

Sequential decision processes (SDPs) were originally identified by Richard Bellman in his pioneering study of *dynamic programming* (DP) [Bellman, 1952, 2010]. Conceptually, SDPs decompose the problem into a set of all possible partial candidate solutions, each such partial solution consisting of a particular combinatorial configuration of the input data. Partial solutions are then combined to construct the optimal, complete solution (this is Bellman’s *principle of optimality*). In Haskell, this process of sequentially constructing partial solutions is expressed as a *recursion* over a *list* of input data:

```
sdp fs e = foldl (choice fs) [e]
choice fs cs x = [f' c' x | f' <- fs, c' <- cs]
```

The function `choice` is implemented as a Haskell *list comprehension* (set comprehension for ordered sequences) which systematically applies each decision function in the given list of decision functions `fs :: [t1 -> t2 -> a]`, to each (partial) input configuration given in the input list `cs :: [t1]`, combined with the next input data `x :: t2`, in the sequence. In these *type declarations*, `<identifier> :: <type string>`, `t1`, `t2`, `a` etc. stand for *arbitrary* data types, the type declaration indicates the *pattern* of types within e.g. the calling structure of functions. These types will be bound to particular data type instances (such as built-in types integer, `Int` or Boolean, `Bool`) when the function is applied to actual data to perform computations. Therefore, each decision function must have type `t1 -> t2 -> a`, indicating that it has two parameters of type `t1` and `t2`, respectively, and outputs a value of type `a`. A list of these is provided, namely, `fs :: [t1 -> t2 -> a]`. By ensuring consistency of among this type information, Haskell can *infer* the type declaration `choice :: [t1 -> t2 -> a] -> [t1] -> t2 -> [a]`.

In turn, this constrains the actual SDP program which applies the decisions in `choice` recursively and has type `sdp :: [t2 -> a -> t2] -> t2 -> [t1] -> [t2]`. The recursion is implemented using Haskell’s built-in (*left*) *list fold* function `foldl :: (b -> a -> b) -> b -> [a] -> b`. Informally, `foldl f e [x1,x2,..,xN]`, computes the value

```
((e 'f' x1) 'f' x2) 'f' .. 'f' xN)
```

where `x 'f' y` indicates *infix* function application as opposed to Haskell’s usual, *prefix* application notation `f x y`, of the two parameter function `f :: b -> a -> b`.

Astute readers may notice that the type definition of the function `sdp` mentions only two parameters `fs` and `e`, but the type declaration mentions another input, the input data list, `[t1]`. This is the use of *currying*: in the more traditional mathematical notation, parameters are *tupled* (x,y) so we write $g(x,y)$, equivalent to Haskell’s `g (x, y)`. But in Haskell we can *fix* parameter `x`—thereby *partially applying* `g`—to obtain the new, single-parameter Haskell function `g x`, taking the lone parameter `y`¹. In this case, this unspecified parameter `[t1]` is inferred from the explicit type of `foldl`.

2.2 An efficient paired combination-sequence generator

The algorithm we will derive for solving the 0-1 loss linear classification problem, will be based on an SDP which pairs size-*D* combinations to enumerate all unique boundary hyperplanes, with linear sequences to evaluate each boundary’s implied 0-1 classification loss. This *combination-sequence generator* will *enumerate* all such pairs of combinatorial configurations, and by evaluating the 0-1 loss over these configurations we will be guaranteed to test every possible assignment, and thus guarantee finding the optimal solution.

The generator for combinations will be based on a generator of *subsequences* given as a special case of function `sdp` defined above,

```
subs = sdp subsfs empty
subsfs = [ignore, include]
ignore c x = c
include c x = c ++ [x]
empty = []
```

where the decisions either “ignore” the new data item `x` or “include” it in the current configuration `c`. To illustrate, for a string (character-type list) input, `subs "abc"` evaluates to `["", "a", "b", "ab", "c", "ac", "ab", "abc"]`.

The generator for linear sequences is trivial having only one decision, and it merely reconstructs the input data,

¹In fact technically *every* function in Haskell only takes one parameter with the successive use of currying like this, the syntax of function declarations mostly hides this fact from the user.

```
seqn = sdp seqnfs empty
seqnfs = [include]
```

such that `seqn "abcd"` evaluates to just `["abcd"]`.

Now, we need to pair together all possible subsequences with linear sequences of the input. This could be done after generating them but only very inefficiently, because every one of the 2^N subsequences would need to be paired with a linear sequence directly. Instead, we can use the *tupling trick* (or “products” [Bird and de Moor, 1997](#)) to perform this pairing during the SDP recursion:

```
sdppair fs1 fs2 e1 e2 = sdp (map paircross (cpp fs1 fs2)) (e1, e2)
pair (f1,f2) x = (f1 x, f2 x)
cross (f1,f2) (a, b) = (f1 a, f2 b)
paircross (f1,f2) = pair . cross (f1,f2)
cpp xs1 xs2 = [(x1', x2') | x1' <- xs1, x2' <- xs2]
```

To understand `sdppair`, first note that the function `cpp :: [a] -> [b] -> [(a, b)]` takes two lists and pairs every element of the first list with every element in the second list. Thus, `cpp subsfs seqnfs` is `[(ignore, include), (include, include)]`, which pairs together the two decision functions of `subs` and `seqn`. The function `paircross` applies the two functions `f1`, `f2` to the first element (`fst`) of its first parameter (which is a tuple), followed by the second parameter (which is a single value), and the second element (`snd`) of the first tuple parameter followed by the second parameter, respectively.² Finally, the built-in Haskell function `map :: (a -> b) -> [a] -> [b]` simply applies this paired update to each of the decision functions in the paired SDPs. In this way, we can now combine the two SDPs into a single SDP:

```
subsseq = sdppair subsfs seqnfs empty empty
```

using which `subsseq "abc"` evaluates to

```
[("", "abc"), ("a", "abc"), ("b", "abc"), ("ab", "abc"), ("c", "abc"),
 ("ac", "abc"), ("bc", "abc"), ("abc", "abc")]
```

The paired SDP generates all subsequences, but we only want subsequences of size D . The simple solution is just to remove the unwanted (*infeasible*) subsequences after they are generated using the built-in Haskell function `filter :: (a -> Bool) -> [a] -> [a]`, which takes a *predicate* `p :: a -> Bool` and deletes any element of the input list where `p` evaluates to `False`. In our case, the predicate

```
fixlen d = (==d) . length . fst
```

with type `Int -> ([a], b) -> Bool` would have the desired effect. Here, this predicate is expressed in the *point-free function composition* style: the function `fst` (picking out the subsequence from each paired SDP configuration) is followed by list `length`, followed by a test for the size being equal to the integer `d` (which is set to D). Using the Haskell composition operator `.` avoids the need to be explicit about the implied configuration input (the “point” in point-free in this case) for this predicate.³

The problem with this simple post-enumeration filtering, is that it is inefficient. Instead, it would be better to filter out any configurations during the SDP recursion, i.e. apply the predicate straight after applying the decision functions. By doing this, we can hope to avoid processing any infeasible configurations. Unfortunately, for `fixlen` this cannot be done because the `subs` SDP constructs subsequences from smaller subsequences generated earlier in the sequence (Bellman’s principle of optimality). However, the slightly modified predicate

```
maxlen d = (<=d) . length . fst
```

has the property we need, preventing subsequences of size larger than D from being retained. Subsequently, once generation is complete, we can remove any subsequences of size $0, 1, 2, \dots, D - 1$ with post-filtering. This is a generic principle: if any feasibility predicate `p` can be written as the conjunction of an auxiliary *prefix-closed* predicate `q` such that `p` and `q` are equal to `p`, then we can *fuse* the SDP with the post-filtering to potentially reduce computational effort:

```
sdppair p q fs1 fs2 e1 e2 = (filter p) .
  foldl (choicefilt q (map paircross (cpp fs1 fs2))) [(e1, e2)]
  choicefilt q fs cs = (filter q) . (choice fs cs)
```

²This might be written as $(f_1, f_2)((c_1, c_2), x) = (f_1(c_1, x), f_2(c_2, x))$ in the usual mathematical notation.

³This point-free notational device is extremely useful for reasoning about algorithms in functional style, and we will make extensive use of it in this paper.

where the new `choicefilt` function simply applies the predicate `q` after applying the decision functions.

This finally leads to our efficient enumerator of paired combination-sequence configurations:

```
combseq d = sdpfiltpair (fixlen d) (maxlen d) subsfs seqnfs empty empty
```

such that, running `combseq 2 "abcd"` gives

```
[("ab", "abcd"), ("ac", "abcd"), ("bc", "abcd"),  
 ("ad", "abcd"), ("bd", "abcd"), ("cd", "abcd")]
```

which to check, agrees with `filter (fixlen 2) (subsseq "abcd")`.

2.3 Exhaustive, incremental cell enumeration (ICE)

As mentioned in the introduction, every hyperplane decision boundary w leads to an implied class assignment $\{+1, -1\}$ for every data point in the dataset, and a corresponding 0-1 loss. There are an uncountable infinity of such boundaries because $w \in \mathbb{R}^{D+1}$ is continuous. Nevertheless, the number of data points N is finite and these points occupy zero volume of the Euclidean feature space \mathbb{R}^{D+1} . This implies that there are a finite number of *equivalence classes* of decision boundaries which share the same assignment. These equivalence classes are Cover's dichotomies, and they form a finite partition of the parameter space \mathbb{R}^{D+1} . It follows that an exact solution for the 0-1 loss classification problem, involves exhaustively enumerating all $O(N^D)$ such dichotomies and then selecting (one of) those with the smallest 0-1 loss. This is the basis on which the combination-sequence algorithm given above, can be used to solve (2) exactly, in polynomial time.

Theorem 1. *Consider a dataset \mathcal{D} of N data points of dimension D in general position, along with their associated labels. All globally optimal solutions to problem (2) for \mathcal{D} , are contained in the set of solutions of all positive and negatively-oriented linear classification decision hyperplanes which go through D out of these N data points.*

Proof. A formal proof is given in the Appendix. □

The informal intuition behind this claim goes as follows. We can construct these solutions by selecting all D out of the N data points, finding the hyperplane which goes exactly through these points, and computing the associated assignments for the entire dataset and their corresponding 0-1 loss. Every such hyperplane has two sets of assignments, corresponding to the two possible orientations of the boundary. However, the D points used to construct these two hyperplanes, have undecided class assignments because the boundary goes exactly through them (so the classification model evaluates to 0 for these points). There are 2^D possible assignments of class labels to the D points on the boundary, and each of these 2^D assignments is a unique dichotomy. The best such dichotomy is the one with the smallest 0-1 loss, and this is guaranteed by selecting the labels of the D points such that they agree with their labels in the training data.⁴

We now have all the ingredients to construct our algorithm, which will enumerate all these linear classification decision hyperplanes and thus solve (2). We will need some basic linear algebra such as real-valued `Vector` and `Matrix` types, solving linear systems `linearsolve :: Matrix -> Vector -> Vector` and matrix-vector multiplication `matvecmult :: Matrix -> Vector -> Vector` which are defined in the imported `Linearsolve` module and listed in the Appendix for completeness.

Dataset

First, the input `Dataset` is defined

```
type Label = Integer  
type Item = (Vector, Label)  
type Dataset = [Item]
```

which is a set of data `Items` which comprises a tuple of a real-valued `Vector` data point and its associated integer training `Label`, for clarity extracted from the tuple using

```
label (x,l) = l  
point (x,l) = x
```

The Haskell keyword `type` simply associates a type with a string label to make code more concise.

⁴We can reason about these geometric facts formally in the original feature space, but the well-developed theory of *hyperplane arrangements* can be used to more clearly structure proofs of these claims through transformation to *dual space* where points are replaced with hyperplanes and vice-versa.

Linear classification

A linear model is the unique hyperplane parameter of type `Vector` which goes through a given set of data points, where the number of data points is equal to the dimension of the space:

```
ones :: Int -> Vector
ones n = take n [1.0,1.0..]

fitw :: Double -> [Vector] -> Vector
fitw sense dx = [-sense] ++ (map (*sense) (linearsolve dx (ones (length (head dx))))))
```

Here, `dx` is a list of vectors of length D , in other words a $D \times D$ matrix, and the function `fitw` solves a linear system of equations to obtain the normal vector of the hyperplane in the homogeneous coordinates for all data in `dx`. The Haskell function `take :: Int -> [a] -> [a]` simply truncates a given list to the given number of elements, and the list comprehension `[1.0,1.0..]` is the infinite list of $+1.0$ real values⁵. The `sense` parameter, taking on the values $\{-1.0, +1.0\}$, is used to select the orientation of the normal vector. The function `head :: [a] -> a` extracts the first element of a list (which must be non-empty); here it is used to find the dimension D of the dataset.

With an (oriented) linear model obtained this way, we can apply it to a set of data points in order to make a decision function prediction

```
evalw :: [Vector] -> Vector -> [Double]
evalw dx w = matvecmult (map ([1.0]++) dx) w
```

which is the oriented distance of all data items in `dx` to the linear model with normal vector `w`. Given that prediction function value, we can obtain the corresponding predicted assignment in $\{0, 1, -1\}$ (which is zero for points which lie on the decision boundary and which actually define the boundary):

```
plabel :: [Double] -> [Label]
plabel = map (round . signum . underflow)
  where
    smalleps = 1e-8
    underflow v = if (abs v) < smalleps then 0 else v
```

The Haskell `where` keyword is a notational convenience which allows local function and variable definitions that can access the enclosing, less indented scope. The reason for the `underflow` correction, is that numerical imprecision leads to predictions for some points which are not exactly on the boundary, where they should be. The function `round` just type casts the label prediction to match the label type (integer). Lastly, combining these two functions above obtains

```
pclass :: [Vector] -> Vector -> [Label]
pclass dx w = plabel (evalw dx w)
```

which, given a set of data points and a hyperplane, obtains the associated labels with respect to that hyperplane.

Loss

Next, given a pair of labels, we want to be able to compute the corresponding term in the 0-1 loss. This makes use of Haskell “case” syntax statements

```
loss01 :: Label -> Label -> Integer
loss01 l1 l2
  | l1 == 0 = 0
  | l2 == 0 = 0
  | l1 /= l2 = 1
  | otherwise = 0
```

This function handles the situation where either label is 0, which occurs for data points which lie on the defining hyperplane and whose predicted class is always assumed to be the same as training label, and also the default case (`otherwise`) to ensure that `loss01` is total. Using this, we can compute the 0-1 loss, E_{0-1} for a given pair of label lists

⁵Haskell is a *lazy* language, in that terms are only evaluated when required. This allows for the specification of non-terminating structures like this, which on evaluation will turn out to be finite.

```
e01 :: [Label] -> [Label] -> Integer
e01 x y = sum (map (\(lx,ly) -> loss01 lx ly) (zip x y))
```

making use of the Haskell function `zip :: [a] -> [b] -> [(a,b)]` which pairs every element of the first given list with the corresponding element of the second given list.

Configuration

Our recursive combination-sequence SDP requires a partial configuration data type which is updated by application of the decisions. For computational efficiency, we package up the linear classification hyperplane defined by the size- D combination, with the 0-1 loss for its corresponding sequence, which together we define as the type (classification) `Model`:

```
type Model = (Vector, Integer)
```

```
modelw :: Model -> Vector
modelw (w,l) = w
```

```
modell :: Model -> Integer
modell (w,l) = l
```

and, combining this with the combination-sequence data type gives us the SDP configuration `Config`:

```
type Config = ([Item], [Item], Maybe Model)
```

```
comb :: Config -> [Item]
comb (c,s,m) = c
```

```
seqn :: Config -> [Item]
seqn (c,s,m) = s
```

```
model :: Config -> Maybe Model
model (c,s,m) = m
```

The first element in this tuple is the combination of data items (with maximum size D) that is used to construct a linear model, the second element is a sequence of data items which have been encountered so far in the recursion. Note that here, the value for `Model` in the configuration is *optional*. This is indicated by the use of Haskell's `Maybe` data type, which is roughly equivalent to allowing variables to take `None` values in Python. Indeed, the initial configuration has empty combination-sequence pairs, and a `Nothing`-valued model:

```
empty :: Config
empty = ([], [], Nothing)
```

The reason for the model pair being optional should be obvious: for combinations of insufficient size, it is not possible to compute a model or corresponding 0-1 loss.

Algorithms

We are now in a position to give the main recursion `e01gen`, which is a combination-sequence SDP:

```
e01gen :: Double -> Integer -> Dataset -> [Config]
e01gen sense ub dxl = foldl (choicefilt retain [cnfupd1,cnfupd2]) [empty] dxl
  where
    dim = length (point (head dxl))
    retain c = (feasible c) && (viable c)
    feasible c = length (comb c) <= dim
    viable c = case (model c) of
      Nothing -> True
      Just (w,l) -> (l <= ub)

    cnfupd1 c xl = case (model c) of
      Nothing -> (comb c, updseqn, Nothing)
```

```

Just (w,l) -> (comb c, updseqn, Just (w, l + e01 [label x1] (pclass [point
  x1] w)))
where
  updseqn = (seqn c) ++ [x1]

cnfupd2 c x1 = case (model c) of
  Nothing -> if (length updcomb == dim) then
    (updcomb, updseqn, Just (w, e01 (map label updseqn) (pclass (map point
      updseqn) w)))
  else
    (updcomb, updseqn, Nothing)
  Just m -> (updcomb, updseqn, Just m)
where
  updcomb = (comb c) ++ [x1]
  updseqn = (seqn c) ++ [x1]
  w = fitw sense (map point updcomb)

```

Given an orientation parameter `sense :: Double`, an approximate upper bound on the 0-1 loss `ub :: Integer`, and a (non-empty) dataset `dx1 :: Dataset`, this outputs a list of candidate solutions to the of type `[Config]` which are potential globally optimal solutions to (2), with 0-1 loss no worse than `ub`. This efficient SDP is derived using all the same principles as section 2.2 above (generic recursive SDPs, tupling, prefix-closed filter fusion), but additionally includes updates to the configurations of type `Config`, and it augments the prefix-closed filter fusion with an additional approximate upper bound predicate. The variable `dim` is just the dimension D of the data points, extracted from the first data item in the dataset.

Regarding configuration updates, the decision function `cnfupd1 :: Config -> Item -> Config`, takes an input configuration, and updates the sequence of the configuration with the new data item. Simultaneously, if there is no defined linear model, it leaves the model unchanged. If there is a model, then the model's 0-1 loss is updated with the loss for the new data item. It leaves the combination part of the configuration unchanged. Conversely, the decision function `cnfupd2 :: Config -> Item -> Config` is responsible for increasing the size of the configuration's combination, therefore, it must also compute a new linear model for the configuration (using `fitw`) when its combination reaches size D for the first time. Thus, the `Maybe` value of the model in the configuration, undergoes a one-way *state transition* from undefined (`Nothing`) to computed (`Just m`); when computed, the linear boundary hyperplane remains unchanged and the configuration's 0-1 loss is updated on each subsequent step of the SDP recursion.

Looking at the filtering, the predicate `feasible :: Config -> Bool` is the function `q`, i.e. `maxlen`, in section 2.2. The predicate `viable :: Config -> Bool` checks whether a linear hyperplane model is defined for a configuration, and if so (case `Just m`), returns `True` when the 0-1 loss of this configuration is at most equal to the approximate upper bound. This is prefix-closed because the 0-1 loss is non-decreasing as more data is scanned by the recursion; this is a very useful computational efficiency improvement since it can eliminate many non-optimal partial solutions. Since the conjunction of two prefix-closed predicates is also prefix-closed, these two predicates are combined, `retain :: Config -> Bool` and form the prefix-closed filtering for the SDP.

Having generated partial solutions, the next stage is to select an optimal one. This involves a straightforward recursive iteration through a non-empty list of partial configurations (`foldl1 :: (a -> a -> a) -> [a] -> a`), comparing adjacent configurations remaining in the list. The best of the pair, that is, one with 0-1 loss at most as large as the other, is selected, using function `best :: Config -> Config -> Config`. At the same time, configurations with `Nothing`-valued (undefined) models are simultaneously removed in the same iteration. This is the implicit application of predicate `def = (/= Nothing) . model`, which satisfies the required filter fusion condition from section 2.2 that `def` and `viable` and `feasible`, equals `def` and `viable`:

```

sel01opt :: [Config] -> Config
sel01opt = foldl1 best
where
  best c1 c2 = case (model c1) of
    Nothing -> c2
    Just (w1,l1) -> case (model c2) of
      Nothing -> c1
      Just (w2,l2) -> if (l1 <= l2) then c1 else c2

```


UCI dataset	N	D	Exact algorithm (ICE)	Support vector machine (SVM)	Logistic regression	Fisher’s linear discriminant
Habermans	306	3	66 (21.6%)	76 (24.8%)	77 (25.2%)	77 (25.1%)
Caesarian	80	5	18 (22.5%)	21 (26.3%)	21 (26.3%)	22 (27.5%)
Cryotherapy	90	6	6 (6.7%)	8 (8.9%)	9 (10.0%)	9 (10.0%)
Acute inflammations	120	6	0 (0.00%)	0 (0.00%)	0 (0.00%)	0 (0.00%)
Blood transfusion	748	4	142 (19.0%)	169 (22.6%)	171 (22.9%)	171 (22.9%)

Table 1: Empirical comparison of the classification error performance E_{0-1} of the exact ICE algorithm developed in this paper, against approximate methods (SVM, logistic regression, linear discriminant) on real-world datasets from the UCI machine learning repository [Dua and Graff, 2019]. Misclassification errors are given as 0-1 loss classification errors, E_{0-1} , and classification error percentage, $\frac{E_{0-1}}{N}$ % in brackets.

Finally, we can give the program for solving problem (2). It generates all positive and negatively-oriented decision boundaries (which are viable with respect to the approximate upper bound ub) and selecting an optimal one:

```
e01class :: Integer -> Dataset -> Config
e01class ub dxy = sel01opt ((e01gen (1.0) ub dxy) ++ (e01gen (-1.0) ub dxy))
```

An approximate upper bound may be computed by any reasonably good approximate method, for instance the support vector machine (SVM). The tighter this bound, the more partial solutions are removed during iteration of `e01gen` which is desirable in order to achieve practical computational performance.

3 Empirical experiments

Various linear classification algorithms were applied to classification data sets from the UCI machine learning repository [Dua and Graff, 2019]. We compare our exact algorithm ICE, against approximate algorithms: support vector machine (SVM) logistic regression and linear discriminant analysis. The results shown in Table 1, clearly demonstrate that ICE will always find solutions with smaller 0-1 loss than approximate algorithms (except for the acute inflammations data set which is linearly separable).

4 Summary, discussion and future work

In this paper, have presented *incremental cell enumeration*, ICE, the first provably correct, worst-case polynomial $O(N^D)$ run-time complexity algorithm for solving the 0-1 loss linear classification problem (2). Our solution combines novel ideas from computational geometry and program transformations in functional programming. Compared with well-known approximate algorithms such as SVMs, logistic regression and Fisher’s linear discriminant, our algorithm is guaranteed to find the best possible solution for a given dataset, where an approximate algorithm cannot be relied upon to find it. Our empirical investigations show that the exact solution is often significantly better than the best approximate solution. From the point of view of interpretability, this is critically important because it demonstrates that, contrary to widely held belief, simple linear models can also be *accurate* i.e. good, purely linear classification models do indeed exist for many practical problems. Approximate algorithms are not guaranteed to be able to find these simple yet accurate linear models, and prior to ICE, provably correct exact algorithms for finding these optimal linear models were computationally intractable. For medium-scale N and small D , we have already demonstrated that ICE can be a good replacement for linear SVMs, for example.

Previous work on this problem of solving (2) is either computationally intractable i.e. worst case exponential time complexity using off-the-shelf mixed-integer programming (MIP) solvers, or is not provably correct [Nguyen and Sanner, 2013]. Previously, we designed and implemented in Python, three other novel, exact algorithms (E01-ICG, E01-ICG-purge, E01-CE) for this problem [Xi and Little, 2023], although correctness proofs and formal algorithm derivations have yet to be provided. These three algorithms have somewhat slower worst-case complexity $O(N^{D+1})$ than ICE. We have performed some small-scale experiments to compare the wall-clock run time of these three algorithms against an earlier branch-and-bound algorithm [Nguyen and Sanner, 2013] and a MIP formulation of the problem solved using GLPK (GNU Linear Programming Kit).

In the future, it would be important to run more sophisticated experiments, comparing highly optimized (C++) implementations of ICE and other such algorithms in terms of practical wall-clock time complexity. Another important future avenue for this research is the development of intrinsically parallel algorithms which would be expected to lead

to much better than $O(N^D)$ worst-case run-time complexity, which may finally make ICE classification practical for problems with large N and D .

References

- Gerald L Alexanderson and John E Wetzel. Simple partitions of space. *Mathematics magazine*, 51(4):220–225, 1978.
- Richard Bellman. On the theory of dynamic programming. *Proceedings of the national Academy of Sciences*, 38(8):716–719, 1952.
- Richard E Bellman. *Dynamic programming*. Princeton university press, 2010.
- Shai Ben-David, Nadav Eiron, and Philip M Long. On the difficulty of approximately maximizing agreements. *Journal of Computer and System Sciences*, 66(3):496–514, 2003.
- Richard Bird and Oege de Moor. *Algebra of Programming*. Prentice Hall, 1997.
- Thomas M Cover. Geometrical and statistical properties of systems of linear inequalities with applications in pattern recognition. *IEEE transactions on electronic computers*, (3):326–334, 1965.
- D. Dua and C. Graff. UCI Machine learning repository, 2019. URL <http://archive.ics.uci.edu/>.
- Herbert Edelsbrunner, Joseph O’Rourke, and Raimund Seidel. Constructing arrangements of lines and hyperplanes with applications. *SIAM Journal on Computing*, 15(2):341–363, 1986.
- Vitaly Feldman, Venkatesan Guruswami, Prasad Raghavendra, and Yi Wu. Agnostic learning of monomials by halfspaces is hard. *SIAM Journal on Computing*, 41(6):1558–1590, 2012.
- Jon Folkman and Jim Lawrence. Oriented matroids. *Journal of Combinatorial Theory, Series B*, 25(2):199–236, 1978.
- Michael McGough. How bad is sacramento’s air, exactly? google results appear at odds with reality, some say. *Sacramento Bee*, 7, 2018.
- Tan Nguyen and Scott Sanner. Algorithms for direct 0–1 loss optimization in binary classification. In *International Conference on Machine Learning*, pages 1085–1093. PMLR, 2013.
- C. Rudin. Why black box machine learning should be avoided for high-stakes decisions, in brief. *Nature Reviews Methods Primers*, 2(1):81, 2022.
- Richard P Stanley et al. An introduction to hyperplane arrangements. *Geometric combinatorics*, 13(389-496):24, 2004.
- Kush R Varshney and Homa Alemzadeh. On the safety of machine learning: Cyber-physical systems, decision sciences, and data products. *Big data*, 5(3):246–255, 2017.
- Rebecca Wexler. When a computer program keeps you in jail: How computers are harming criminal justice. *New York Times*, 13, 2017.
- He Xi and Max A. Little. Exact 0-1 loss linear classification algorithms, April 2023. URL <https://github.com/XiHegrt/E01Loss>.

A Algorithm correctness (proof of Theorem 1)

Before we prove this, we need to introduce a few definitions and prove some lemmas. Here, we introduce the *dual transformation* converting a configuration of data points into a *hyperplane arrangement*. A *point configuration* is a set of data points $\mathbf{p}_n \in \mathbb{R}^D$, $n \in \mathcal{N} = \{1, \dots, N\}$, where D is the dimension of a (generally affine) *primal space*. We denote this point set by \mathcal{P} . A finite *hyperplane arrangement* $\mathcal{H} = \{\mathbf{h}_1, \dots, \mathbf{h}_N\}$ is a finite set of hyperplanes $\mathbf{h}_n = \{\mathbf{v} \in \mathbb{R}^D : \boldsymbol{\alpha}^T \mathbf{v} = c\}$ in \mathbb{R}^D for some constant $c \in \mathbb{R}$. We call \mathcal{H} a *central arrangement*, if $c = 0$. A point configuration or hyperplane arrangement in *general position* is called *simple*; simple arrangements are defined below.

Definition 1. *General position.* A set of points in D -dimensional affine space is in *general position* if no k of them lie in a $(k - 2)$ -dimensional linear subspace of \mathbb{R}^D (such linear subspaces are called *flats*; flats of $D - 1$ dimension are called *hyperplanes*), for $k = 2, 3, \dots, D + 1$. An arrangement is in general position if

$$\{\mathbf{h}_1, \dots, \mathbf{h}_k\} \in \mathcal{H}, k \leq D \implies \dim(\mathbf{h}_1 \cap \dots \cap \mathbf{h}_k) = D - k, \quad (3)$$

where $\dim(R)$ is the *dimension* of the space R and

$$\{\mathbf{h}_1, \dots, \mathbf{h}_k\} \in \mathcal{H}, k > D \implies \mathbf{h}_1 \cap \dots \cap \mathbf{h}_k = \{\emptyset\}. \quad (4)$$

In words, \mathcal{H} is in general position if the intersection of any k hyperplanes is contained in a $(D - k)$ -dimensional *flat*, for $1 \leq k \leq D$. For example, if $D = 2$ then a set of lines is in general position if no two are parallel and no three meet at a point. We call an arrangement in general position a *simple arrangement*.

Definition 2. *Faces of a hyperplane arrangement.* We denote with $\mathbf{h}_n^+ = \{\mathbf{v} \in \mathbb{R}^D : \boldsymbol{\alpha}^T \mathbf{v} > c\}$, $\mathbf{h}_n^- = \{\mathbf{v} \in \mathbb{R}^D : \boldsymbol{\alpha}^T \mathbf{v} < c\}$ the open half-spaces “above” or “below” h_n . We associate to each point $\mathbf{x} \in \mathbb{R}^D$, a *sign vector* $\delta(\mathbf{x})$ with respect to arrangement \mathcal{H} , defined by

$$\delta_n(\mathbf{x}) = \begin{cases} +1 & \mathbf{x} \in \mathbf{h}_n^+ \\ 0 & \mathbf{x} \in \mathbf{h}_n \\ -1 & \mathbf{x} \in \mathbf{h}_n^- \end{cases} \quad (5)$$

which first calculates the distance of \mathbf{x} to hyperplane \mathbf{h}_n , and then finds the sign of this distance. We denote by $\mathcal{F}_{\mathcal{H}}$ the set of sign vectors $\delta(\mathbf{x})$,

$$\mathcal{F}_{\mathcal{H}} = \{\delta(\mathbf{x}) : \mathbf{x} \in \mathbb{R}^D\}, \quad (6)$$

Using this set, we can define a *face* \mathbf{f} of the arrangement as the subset of \mathbb{R}^D represented by $\mathbf{f} = \bigcap_{n \in \mathcal{N}} \mathbf{h}_n^{\delta(\mathbf{f})}$ for the sign vector $\delta(\mathbf{f}) \in \mathcal{F}_{\mathcal{H}}$. In fact, \mathbf{f} defines an equivalence class in \mathbb{R}^D . Different vectors in $\mathcal{F}_{\mathcal{H}}$ define different faces of the arrangement \mathcal{H} . We call a face *k-dimensional* if it is contained in a *k-flat* for $-1 \leq k \leq D + 1$. Some special faces are given the name *vertices* ($k = 0$), *edges* ($k = 1$), and *cells* ($k = D$). A k -face \mathbf{g} and a $(k - 1)$ -face \mathbf{f} are said to be *incident* if \mathbf{f} is contained in the *closure* of face \mathbf{g} , for $1 \leq k \leq D$. In that case, face \mathbf{g} is called a *superface* of \mathbf{f} , and \mathbf{f} is called a *subface* of \mathbf{g} .

Definition 3. *Primal and dual spaces.* Let $\mathbf{p} = (p_1, p_2, \dots, p_D)^T \in \mathbb{R}^D$. The geometric *dual transformation* $\phi : \mathbb{R}^D \rightarrow \mathbb{R}^D$ maps a point \mathbf{p} to a hyperplane (a subspace of \mathbb{R}^D), $\phi(\mathbf{p})$, given by the equation

$$x_D = p_1 x_1 + p_2 x_2 + \dots + p_{D-1} x_{D-1} - p_D, \quad (7)$$

and conversely, the function ϕ^{-1} transforms a (non-vertical) hyperplane $\mathbf{h} : x_D = h_1 x_1 + h_2 x_2 + \dots + h_{D-1} x_{D-1} - h_D$ to a point $\phi^{-1}(\mathbf{h}) = (p_1, p_2, \dots, p_D)^T$. These functions are inverses, so that $\phi^{-1}(\phi(\mathbf{p})) = \mathbf{p}$. This transformation is naturally extended to sets of points, $\phi(\mathcal{P})$ and sets of hyperplanes, $\phi(\mathcal{H})$. We use the terms *primal space*, and *dual space* to spaces before and after transformation by ϕ and ϕ^{-1} .

Theorem 2. Let \mathbf{p} be a point and \mathbf{h} be a hyperplane in \mathbb{R}^D . Under the dual transformation ϕ , \mathbf{p} and \mathbf{h} satisfy the following properties:

1. *Incidence preservation:* point \mathbf{p} belongs to hyperplane \mathbf{h} if and only if point $\phi^{-1}(\mathbf{h})$ belongs to hyperplane $\phi(\mathbf{p})$,
2. *Order preservation:* point \mathbf{p} lies above (below) hyperplane \mathbf{h} if and only if point $\phi^{-1}(\mathbf{h})$ lies above (below) hyperplane $\phi(\mathbf{p})$.

Lemma 1. For a data set \mathcal{D} in general position, each of Cover’s dichotomies corresponds to a cell in the dual space, and dichotomies corresponding to bounded cells have no complement cell (cells with reverse sign vector). Dichotomies corresponding to the unbounded cells in the dual arrangements $\phi(\mathcal{D})$ have a complement cell.

Proof. The first statement is true because of the order preservation property—data item \mathbf{x} lies above (below) hyperplane \mathbf{h} if and only if point $\phi^{-1}(\mathbf{h})$ lies above (below) hyperplane $\phi(\mathbf{x})$. For a data set \mathcal{D} and hyperplane \mathbf{h} , assume \mathbf{h} has a normal vector \mathbf{w} (in homogeneous coordinates) and there is no data item lying on \mathbf{h} . Then, hyperplane \mathbf{h} will partition the set \mathcal{D} into two subsets $\mathcal{D}^+ = \{\mathbf{x}_n : \mathbf{w}^T \mathbf{x} > 0\}$ and $\mathcal{D}^- = \{\mathbf{x}_n : \mathbf{w}^T \mathbf{x} < 0\}$, and according to the Theorem 2, \mathcal{D} has a unique associated dual arrangement $\phi(\mathcal{D})$. Thus, the sign vector of the point $\phi^{-1}(\mathbf{h})$ with respect to arrangement $\phi(\mathcal{D})$ partitions the arrangement into two subsets $\phi(\mathcal{D})^+ = \{\phi(\mathbf{x}_n) : \boldsymbol{\nu}_{\phi(\mathbf{x}_n)}^T \phi^{-1}(\mathbf{h}) > 0\}$ and $\phi(\mathcal{D})^- = \{\boldsymbol{\nu}_{\phi(\mathbf{x}_i)}^T \phi^{-1}(\mathbf{h}) < 0\}$, where $\boldsymbol{\nu}_{\phi(\mathbf{x}_n)}^T$ is the normal vector to the dual hyperplane $\phi(\mathbf{x}_n)$.

Next, we need to prove that bounded cells have no complement cell. The reverse assignment of the bounded cells of the dual arrangements $\phi(\mathcal{D})$ cannot appear in the primal space since the transformation ϕ can only have normal vector $\boldsymbol{\nu}$ pointing in one direction, in other words, transformation $\phi: x_D = p_1x_1 + p_2x_2 + \dots + p_{D-1}x_{D-1} - p_D$ implies the D -th component of normal vector $\boldsymbol{\nu}$ is -1 . For unbounded cells, in dual space, every unbounded cell \boldsymbol{f} associates with another cell \boldsymbol{g} , such that \boldsymbol{g} has opposite sign vector to \boldsymbol{f} . This is because every hyperplane $\phi(\boldsymbol{x}_n)$ is cut by another $N - 1$ hyperplanes into $N + 1$ pieces (since in a simple arrangement no two hyperplanes are parallel), and each of the hyperplanes contains two rays, call them $\boldsymbol{r}_1, \boldsymbol{r}_2$. These two rays point in opposite directions, which means that the cell incident with \boldsymbol{r}_1 has an opposite sign vector to \boldsymbol{r}_2 with respect to all other $N - 1$ hyperplanes. Therefore, we only need to take the cell \boldsymbol{f} incident with \boldsymbol{r}_1 , and in the positive direction with respect to $\phi(\boldsymbol{x}_n)$, take \boldsymbol{g} to be the cell incident with \boldsymbol{r}_2 , and in the negative direction with respect to $\phi(\boldsymbol{x}_n)$. In this way, we obtain two unbounded cells which have opposite sign vector. This means that, for these unbounded cells, we can move the position of hyperplane \boldsymbol{h} and get the opposite assignment of set \mathcal{P} . This corresponds, in the dual space, to moving a point \boldsymbol{p} inside the cell \boldsymbol{f} , to cell \boldsymbol{g} . \square

Lemma 2. *For a data set \mathcal{D} in general position, a hyperplane with d data items lies on it, $0 \leq d \leq D$ correspond to a $(D - d)$ -face in the dual arrangement $\phi(\mathcal{D})$. Hyperplanes with D points lying on it, corresponding to vertices in the dual arrangement.*

Proof. According to the incidence preservation property, d data items lying on a hyperplane will intersect with d hyperplanes, and the intersection of d hyperplanes will create a $(D - d)$ -dimensional space, which is a $(D - d)$ -face, and the 0-faces are the vertices of the arrangement. \square

Lemma 3. *For a data point set $\mathcal{P} \in \mathbb{R}^D$, the total number of linearly separable assignments in Cover's function counting theorem, is the same as the number of cells of the dual arrangements $\phi(\mathcal{P})$, plus the number of bounded cells of the dual arrangements $\phi(\mathcal{P})$.*

Proof. Lemma 1 show that the bounded cells can only obtain one linearly separable assignment per dichotomy, and the dichotomies corresponding to the unbounded cells can have a pair of dual assignments. Therefore, if we reverse the sign vector of bounded cells and combine with the assignment in each cell, we obtain all linearly separable assignments in Cover's function counting theorem. \square

Remark. Assuming we have linear hyperplanes in Cover's function counting theorem, when we apply Cover's function counting theorem, we need to embed our data set in homogeneous space (representing our data points in homogeneous coordinates). Therefore, according to Cover's counting theorem, for N data points in D -dimensions, the number of linearly separable assignments is

$$C(N, D + 1) = 2 \sum_{d=0}^D \binom{N-1}{d}. \quad (8)$$

Alexanderson and Wetzel [1978] (and see also Edelsbrunner et al., 1986) show that, for a set of $(D - 1)$ -dimensional hyperplanes $\mathcal{H} = \{\boldsymbol{h}_1, \dots, \boldsymbol{h}_N\}$, the number of cells is

$$C_D(\mathcal{H}) = \sum_{d=0}^D \binom{N}{d}. \quad (9)$$

Also, Stanley et al. [2004] shows that the number of bounded regions is

$$B_D(\mathcal{H}) = \binom{N-1}{D}. \quad (10)$$

Putting this information together, we obtain

$$\begin{aligned}
B_D(\mathcal{H}) + C_D(\mathcal{H}) &= \binom{N-1}{D} + \sum_{d=0}^D \binom{N}{d} \\
&= \binom{N-1}{D} + \sum_{d=0}^D \left[\binom{N-1}{d} + \binom{N-1}{d-1} \right] \\
&= \sum_{d=0}^D \binom{N-1}{d} + \sum_{i=0}^D \binom{N-1}{d} \\
&= 2 \sum_{d=0}^D \binom{N-1}{d} \\
&= C(N, D+1).
\end{aligned} \tag{11}$$

Definition 4. *Separation set.* The separation set $S(\mathbf{f}, \mathbf{g})$ for two faces \mathbf{f}, \mathbf{g} is defined by

$$S(\mathbf{f}, \mathbf{g}) = \{n \in \mathcal{N} : \delta_n(\mathbf{f}) = -\delta_n(\mathbf{g}) \neq 0\}, \tag{12}$$

using which, we say that the two faces \mathbf{f}, \mathbf{g} are *conformal* if $S(\mathbf{f}, \mathbf{g}) = \emptyset$.

Two faces that are conformal is essentially the same as saying that two faces has consistent assignments.

Lemma 4. *Two faces \mathbf{f}, \mathbf{g} are conformal if and only if \mathbf{f} and \mathbf{g} are subfaces of a common proper cell or one face is a subface of the other.*

A similar result is described in oriented matroid theory [Folkman and Lawrence, 1978].

Lemma 5. *For an arbitrary maximal face (cell) \mathbf{f} , the sign vector of \mathbf{f} is $\delta(\mathbf{f})$. For an arbitrary $(D-d)$ -dimension face \mathbf{g} , $0 < d \leq D$, the number of different signs of $\delta(\mathbf{g})$ with respect to $\delta(\mathbf{f})$ (the 0-1 loss of $\delta(\mathbf{g})$ with respect to $\delta(\mathbf{f})$) is larger than or equal to D , where equality holds only when \mathbf{g} is conformal to \mathbf{f} (\mathbf{g} is a subface of \mathbf{f}).*

Proof. In a simple arrangement, the sign vector $\delta(\mathbf{f})$ of a cell \mathbf{f} has no zero signs, and a $(D-d)$ -dimension face has d zero signs. Assume $\delta(\mathbf{f}) = \mathbf{l}$ (i.e. let $\delta(\mathbf{f})$ be the true label vector), then $E_{0-1}(\delta(\mathbf{g})) \geq d$. If the $S(\mathbf{f}, \mathbf{g}) = \emptyset$, $E_{0-1}(\delta(\mathbf{g})) = d$, hence \mathbf{g} is conformal to \mathbf{f} . If $S(\mathbf{f}, \mathbf{g}) \neq \emptyset$, and assuming $|S(\mathbf{f}, \mathbf{g})| = C$, according to the definition of the objective function and conformal faces, $E_{0-1}(\delta(\mathbf{g})) = d + C$. Hence, $E_{0-1}(\delta(\mathbf{g})) \geq d$, and equality holds only when \mathbf{g} is conformal to \mathbf{f} . \square

We are now ready to prove Theorem 1, which we repeat here for ease of reference.

Theorem. *Consider a dataset \mathcal{D} of N data points of dimension D in general position, along with their associated labels. All globally optimal solutions to problem (2), are equivalent (in terms of 0-1 loss) to the optimal solutions contained in the set of solutions of all positive and negatively-oriented linear classification decision hyperplanes which go through D out of N data points in the dataset \mathcal{D} .*

Proof. First, we transform a dataset \mathcal{D} to its dual arrangement, according to Lemma 1 and Lemma 1, each dichotomy has a corresponding dual cell, if we evaluate the sign vectors for all possible cells in the dual arrangement and their reverse signs, we can obtain the optimal solution for the 0-1 loss classification problem. Assume the optimal cell is \mathbf{f} , we need to prove that, one of the adjacent vertex for this cell is also the optimal vertex. Then the finding a optimal vertex is equivalent to finding a optimal cell since the optimal cell is one of the adjacent cell of this vertex. According to Lemma 5, any vertexes that are non-conformal to \mathbf{f} , its 0-1 loss with respect to $\delta(\mathbf{f})$ will strictly greater than D . Since \mathbf{f} is optimal, any sign vectors with larger sign difference (with respect to $\delta(\mathbf{f})$) will have larger 0-1 loss value (with respect to true label \mathbf{l}). Therefore, vertices that are conformal to \mathbf{f} will have smaller 0-1 loss value, thus we can evaluate all vertices (and the reverse sign vector for these vertices) and choose the best one, which, according to Lemma 2, is equivalent to evaluating all possible positive and negatively-oriented linear classification decision hyperplanes and choosing one linear decision boundary with the smallest 0-1 loss value. \square

B Haskell linear algebra functions

Here we list the `Linearsolve` module functions used to perform the linear algebra for the ICE algorithm.

```

module LinearSolve
  (vecmult, matvecmult, matmult, linearsolve, triangular, rotatePivot,
   resubstitute, resubstitute', Matrix, Vector)
where
import Data.List
type Vector = [Double]
type Row    = [Double]
type Matrix = [Row]

vecmult :: Vector -> Vector -> Double
vecmult a b = sum (zipWith (*) a b)

matvecmult :: Matrix -> Vector -> Vector
matvecmult rows v = [sum (zipWith (*) row v) | row <- rows]

matmult :: Matrix -> Matrix -> Matrix
matmult a b
  | null a || null b = []
  | length (head a) /= length b = error "Incompatible matrices for multiplication"
  | otherwise = [[sum (zipWith (*) row col) | col <- (transpose b)] | row <- a]

linearsolve :: Matrix -> Vector -> Vector
linearsolve a b = x
  where
    b' = map (\y -> [y]) b
    a' = zipWith (++) a b'
    x = resubstitute $ triangular a'

triangular :: Matrix -> Matrix
triangular [] = []
triangular m = row:(triangular rows')
  where
    (row:rows) = rotatePivot m
    rows' = map f rows
    f bs
      | (head bs) == 0 = drop 1 bs
      | otherwise      = drop 1 $ zipWith (-) (map (*c) bs) row
      where
        c = (head row)/(head bs)

rotatePivot :: Matrix -> Matrix
rotatePivot (row:rows)
  | (head row) /= 0 = (row:rows)
  | otherwise      = rotatePivot (rows ++ [row])

resubstitute :: Matrix -> Vector
resubstitute = reverse . resubstitute' . reverse . map reverse

resubstitute' :: Matrix -> Vector
resubstitute' [] = []
resubstitute' (row:rows) = x:(resubstitute' rows')

```

```
where
  x      = (head row)/(last row)
  rows' = map substituteUnknown rows
  substituteUnknown (a1:(a2:as')) = ((a1-x*a2):as')
```